Laboratorní cvičení - Diferenciální rovnice

Směrové pole

Pro nakreslení směrového pole slouží příkaz **dfieldplot(ODE, y(x), x=a..b, y=c..d, opts),** kde **ODE**je daná diferenciální rovnice, **y(x)**je název hledané funkce, **x=a..b, y=c..d**jsou rozsahy na osách a **opts**jsou nepovinné parametry. Před použitím tohoto příkazu je potřeba načíst balíček **DEtools**.

Příklad 1. Nakreslete směrové pole diferenciální rovnice $y' = x^2 + y^2$.

```
> with(DEtools):
> DR:=diff(y(x),x)=x^2+y(x)^2;
```

$$DR := \frac{d}{dx} y(x) = x^2 + y(x)^2$$

```
> dfieldplot(DR,y(x),x=-3..3,y=-3..3,dirgrid=[20,20]);
```



Při kreslení jsme využili nepovinný parametr **dirgrid=[m, n]**, který nastavuje počet bodů sítě. Směrové pole pak tvoří
m x n segmentů.

Příklad 2. Znázorněte integrální křivky diferenciální rovnice z Příkladu 1., které splňují podmínky $y(0) = 2, y(0) = 0,\ y(0) = -1.$

Pro nakreslení integrálních křivek použijeme příkaz **DEplot(ODE, y(x), x=a..b, [[PP1],[PP2],...], opts),** kde **ODE**je daná diferenciální rovnice, **y(x)**je symbol pro řešení,

x=a..bje rozsah nezávisle proměnné a **PP1, PP2**jsou počáteční podmínky, opts jsou další parametry. Příkaz DEplot je součástí balíčku DEtools. Před jeho prvním použitím je tedy třeba balíček načíst příkazem **with(DEtools).**

```
> DEplot(DR,y(x),x=-3..3,y=-3..3,[[y(0)=0], [y(0)=2],
[y(0)=-1]],linecolor=[blue,green,gold]);
```



Jako nepovinný parametr zde byl použit rozsah pro y-ovou souřadnici a barvy integrálních křivek.

Příklad 3. Nakreslete směrové pole diferenciální rovnice $y' = \dfrac{1}{x+2y}$ a znázorněte integrální křivky splňující podmínky

$y(1)=0, y(0)=2, y(-3)=0.$

```
> restart;
> with(DEtools):
> DR:=diff(y(x),x)=1/(x+2*y(x));
```

$$DR := \frac{d}{dx}\, y(x) = \frac{1}{x+2\,y(x)}$$

```
> DEplot([DR],[y(x)],x=-3..3,[[y(1)=0], [y(0)=2],
[y(-3)=0]],y=-1..3, stepsize=.01, linecolor=[blue,green,gold],
scaling=constrained);
Warning, plot may be incomplete, the following errors(s) were issued:
   cannot evaluate the solution further left of .81093019, probably
a singularity
```

```
Warning, plot may be incomplete, the following errors(s) were issued:
   cannot evaluate the solution further left of -1.8027758, probably
a singularity
```



Pravá strana dané diferenciální rovnice není definovaná v bodech, které leží na přímce o rovnici $x + 2y = 0$. Zelené a modré řešení tudíž nelze prodloužit více směrem doleva. Není definované na celém požadovaném rozsahu pro x. To je důvodem, proč Maple vypsal dvě varování. Z nich je vidět, kam až doleva jsou příslušná řešení definována.

Diferenciální rovnice 1. řádu

Pro řešení diferenciálních rovnic použijeme příkazu **dsolve(ODE, y(x), opts),** kde **ODE**je daná diferenciální rovnice,
y(x)je název hledané funkce a **opts**jsou nepovinné parametry.
POZOR! Pokud příkaz **dsolve** nedá výsledek, ani nevypíše chybové hlášení, je potřeba úlohu řešit numerickou metodou.
Derivaci funkce *y* podle *x* zapíšeme příkazem **diff(y(x), x).** Je-li pravá strana rovnice rovna nule, nemusíme ji psát. Integrační konstanta je v systému Maple značena **_C1**. Chceme-li najít řešení počáteční úlohy, použijeme příkaz **dsolve({ODE, PP}, y(x))**, kde parametr **PP**je počáteční podmínka.
Příklad 1. Nalezněte obecné řešení diferenciální rovnice
$\frac{y}{y'} - x = 0$, dále pak nalezněte řešení, která vyhovují počátečním podmínkám a) y(4)=1 , b) y(4)=3.

```
> restart;
> DR:=y(x)/diff(y(x),x)-x=0;
```

$$DR := \frac{y(x)}{\frac{d}{dx}\,y(x)} - x = 0$$

```
> dsolve(DR,y(x));
```

$$y(x) = \_C1\,x$$

Řešením jsou v našem případě přímky procházející počátkem.
Pokud nechceme opakovat postup dosazování počátečních podmínek do příkazu **dsolve({ODE, PP}, y(x))**, je možné podmínky zapsat jako seznam, tj. do hranatých závorek, a odvolávat se na jejich pořadí.

```
> PP:=[y(4)=1,y(4)=3];
```

$$PP := [y(4) = 1, y(4) = 3]$$

```
> dsolve({DR,PP[1]}, y(x));
```

$$y(x) = \frac{1}{4}\,x$$

```
> dsolve({DR,PP[2]}, y(x));
```

$$y(x) = \frac{3}{4}\,x$$

Příklad 2. Nakreslete směrové pole diferenciální rovnice $x\,y' + (x+1)\,y = 2\,xe^{-x}$. Nalezněte řešení, která vyhovují počáteční podmínce $y(1) = a$.

```
> restart;
> with(DEtools):
> DR:=x*diff(y(x),x)+(x+1)*y(x)=2*x*exp(-x);
```

$$DR := x\left(\frac{d}{dx}\,y(x)\right) + (x+1)\,y(x) = 2\,x\,e^{-x}$$

```
> dfieldplot(DR,y(x),x=-1..4,y=-2..2,scaling=constrained);
```



Zadání počáteční podmínky:

```
> PP:=y(1)=a;
```

$$PP := y(1) = a$$

Řešení DR s danou počáteční podmínkou

```
> res:=dsolve({DR,PP},y(x));
```

$$res := y(x) = \frac{\left(x^2 - \frac{-a + e^{-1}}{e^{-1}}\right) e^{-x}}{x}$$

Pro každou hodnotu a máme jiné řešení. Volbou parametru *a* dostaneme různé integrální křivky. Volme např.

y(1)=0,2; y(1)= -0.2; y(1)=1; y(1)= -1. Počáteční podmínky zadáme tentokrát přímo do příkazu DEplot.

```
>
DEplot(DR,y(x),x=-1..4,y=-2..2,[[y(1)=0.2],[y(1)=-0.2],[y(1)=1],
[y(1)=-1]],linecolor=[blue,green,gold,red],scaling=constrained);
```



Diferenciální rovnice vyšších řádů

Pro řešení diferenciálních rovnic vyšších řádů použijeme příkazu **dsolve(ODE, y(x), opts),** kde **ODE**je daná diferenciální rovnice,
y(x)je název hledané funkce a **opts**jsou nepovinné parametry. Vyšší derivaci funkce *y* podle *x* zapíšeme příkazem
diff(y(x), x$n), kde n je stupeň diferenciální rovnice. Je-li pravá strana rovnice rovna nule, nemusíme ji psát. Integrační konstanty jsou v systému Maple značeny **_C1, _C2** atd. Chceme-li najít řešení počáteční úlohy, použijeme příkaz **dsolve({ODE, PP}, y(x))**, kde parametr **PP**jsou počáteční podmínky.

Příklad 1. Najděte řešení homogenní lineární diferenciální rovnice $y''' - 3\,y'' - 2\,y' = 0.$

```
> DR:=diff(y(x),x$3)-3*diff(y(x),x$2)+2*diff(y(x),x);
```

$$DR := \frac{d^3}{dx^3}\,y(x) - 3\left(\frac{d^2}{dx^2}\,y(x)\right) + 2\left(\frac{d}{dx}\,y(x)\right)$$

```
> dsolve(DR,y(x));
```

$$y(x) = \_C1 + \_C2\,e^x + \_C3\,e^{2\,x}$$

Příklad 2. Najděte řešení nehomogenní lineární diferenciální rovnice $2\,y'' + 8\,y = \frac{1}{\sin^3(2\,x)}$ s počátečními podmínkami $y\left(\frac{\pi}{4}\right) = \frac{15}{16}, y'\left(\frac{\pi}{4}\right) = 4.$

```
> DR:=2*diff(y(x),x$2)+8*y(x)=1/sin(2*x)^3;
```

$$DR := 2\left(\frac{d^2}{dx^2}\,y(x)\right) + 8\,y(x) = \frac{1}{\sin(2\,x)^3}$$

```
> dsolve(DR,y(x));
```

$$y(x) = \sin(2\,x)\,\_C2 + \cos(2\,x)\,\_C1 + \frac{1}{16}\,\frac{-1 + 2\cos(2\,x)^2}{\sin(2\,x)}$$

Počáteční podmínky lze zapsat společně do proměnné PP.Derivaci funkce v daném bodě zapíšeme pomocí **D(y)(x_0).**

```
> PP:=y(Pi/4)=15/16, D(y)(Pi/4)=4;
```

$$PP := y\left(\frac{1}{4}\,\pi\right) = \frac{15}{16},\ D(y)\left(\frac{1}{4}\,\pi\right) = 4$$

```
> dsolve({DR,PP},y(x));
```

$$y(x) = \sin(2\,x) - 2\cos(2\,x) + \frac{1}{16}\,\frac{-1 + 2\cos(2\,x)^2}{\sin(2\,x)}$$

Příklad 3. Najděte řešení lineární diferenciální rovnice druhého řádu se speciální pravou stranou $y'' - 6\,y' + 9\,y = 2\,x^2 - x + 3$.

```
> DR:=diff(y(x),x$2)-6*diff(y(x),x)+9*y(x)=2*x^2-x+3;
```

$$DR := \frac{d^2}{dx^2}\,y(x) - 6\left(\frac{d}{dx}\,y(x)\right) + 9\,y(x) = 2\,x^2 - x + 3$$

```
> dsolve(DR,y(x));
```

$$y(x) = e^{3\,x}\,\_C2 + e^{3\,x}\,x\,\_C1 + \frac{11}{27} + \frac{5}{27}\,x + \frac{2}{9}\,x^2$$

Příklad 4.Najděte řešení lineární diferenciální rovnice druhého řádu $y'' - y' = 2 - 2\,x$ s počátečními podmínkami y(0)=0, y'(0)=1.

```
> restart;
> with(DEtools):
> DR:= diff(y(x),x$2)-diff(y(x),x)=2*(1-x);
```

$$DR := \frac{d^2}{dx^2}\,y(x) - \left(\frac{d}{dx}\,y(x)\right) = 2 - 2\,x$$

Počáteční podmínky lze zapsat společně do proměnné PP.Derivaci funkce v daném bodě zapíšeme pomocí **D(y)(x_0).**

```
> PP:=y(0)=0,D(y)(0)=1;
```

$$PP := y(0) = 0,\ D(y)(0) = 1$$

```
> res:=dsolve({DR,PP},y(x));
```

$$res := y(x) = x^2 + e^x - 1$$

Ze získaného řešení vyrobíme příkazem **unapply**funkci proměnné *x*. V zápise je potřeba se odkázat pouze na pravou stranu funkce res, což uděláme pomocí příkazu **rhs** (right hand side).

```
> f:=unapply(rhs(res),x);
```

$$f := x \to x^2 + e^x - 1$$

Funkci *f*,a tím pádem i řešení diferenciální rovnice, si můžeme pomocí příkazu plot vykreslit.

```
> plot(f(x),x=-2..2,y=-1..1);
```



Příklad 5. Najděte řešení lineární diferenciální rovnice třetího řádu $y''' + 2y'' + y' = -2xe^{-2x}$ s počátečními podmínkami $y(0) = 2, y'(0) = 1, y''(0) = 0.$

```
> restart;
>
DR:=diff(y(x),x$3)+2*diff(y(x),x$2)+diff(y(x),x)=-2*x*exp(-2*x);
```

$$DR := \frac{d^3}{dx^3}\, y(x) + 2\left(\frac{d^2}{dx^2}\, y(x)\right) + \frac{d}{dx}\, y(x) = -2\, x\, e^{-2\,x}$$

```
> dsolve(DR,y(x));
```

$$y(x) = -e^{-x}\_C2 + \_C1\left(-e^{-x}x - e^{-x}\right) + \frac{5}{2}\, e^{-2\,x} + x\, e^{-2\,x} + \_C3$$

Všimněte si zápisu počáteční podmínky pro druhou derivaci (je též možné použít zápis (D@@2)(y)(0)).

```
> PP:=y(0)=2,D(y)(0)=1,D(D(y))(0)=0;
```

$$PP := y(0) = 2, \mathrm{D}(y)(0) = 1, \mathrm{D}^{(2)}(y)(0) = 0$$

```
> res:=dsolve({DR,PP},y(x));
```

$$res := y(x) = -4\, e^{-x} + e^{-x}x + \frac{5}{2}\, e^{-2\,x} + x\, e^{-2\,x} + \frac{7}{2}$$

```
> f:=unapply(rhs(combine(res)),x);
```

$$f := x \to -4\, e^{-x} + e^{-x}x + \frac{5}{2}\, e^{-2\,x} + x\, e^{-2\,x} + \frac{7}{2}$$

```
> plot(f(x),x=-2..2,y=-2..4);
```




```
> restart;
```

Soustavy diferenciálních rovnic

Příklad 1. Nalezněte řešení systému

$y_1' = y_1 + y_2$,

$y_2' = 4\,y_1 + y_2$.

Systém není potřeba zapisovat po jednotlivých rovnicích, ale můžeme je zapsat do jednoho přiřazení a oddělit je čárkami.

```
> sys_DR:=diff(y_1(x),x)=y_1(x)+y_2(x),
diff(y_2(x),x)=4*y_1(x)+y_2(x);
```

$$sys\_DR := \frac{d}{dx}\,y\_1(x) = y\_1(x) + y\_2(x),\ \frac{d}{dx}\,y\_2(x) = 4\,y\_1(x) + y\_2(x)$$

Řešení systému lze napsat následovně:

```
> dsolve([sys_DR]);
```

$$\{y\_1(x) = \_C1\,e^{-x} + \_C2\,e^{3\,x},\ y\_2(x) = -2\,\_C1\,e^{-x} + 2\,\_C2\,e^{3\,x}\}$$

Příklad 2. Nalezněte řešení systému

$y_1' = 3\,y_1 + 6\,y_2$,

$y_2' = -y_1 - 2\,y_2$.

s počáteční podmínkou $y_1(0) = 2, y_2(0) = 1$.

```
> sys_DR:=diff(y_1(x),x)=3*y_1(x)+6*y_2(x),
diff(y_2(x),x)=-y_1(x)-2*y_2(x);
```

$$sys\_DR := \frac{d}{dx} y\_1(x) = 3\, y\_1(x) + 6\, y\_2(x), \frac{d}{dx} y\_2(x) = -y\_1(x) - 2\, y\_2(x)$$

Počáteční podmínky zapíšeme společně do proměnné PP.

```
> PP:=y_1(0)=2, y_2(0)=1;
```

$$PP := y\_1(0) = 2, y\_2(0) = 1$$

V zápise řešení je potřeba hledané funkce y_1, y_2 zapsat jako seznam, tj. do hranatých závorek.

```
> dsolve({sys_DR,PP},[y_1(x), y_2(x)]);
```

$$\{y\_1(x) = 12\, e^x - 10, y\_2(x) = 5 - 4\, e^x\}$$

Numerické řešení soustav diferenciálních rovnic - metoda Rungeho - Kutty, Eulerova metoda

Pokud řešíme diferenciální rovnice numericky, je potřeba zadat volbu **numeric** a případně metodu, která se má použít (implicitně je dáno **method=rkf45)**. Výstupem je procedura pro výpočet hodnoty řešení v bodě x **.**

Příklad 1. Řešte diferenciální rovnici $y'y^2 + y = \sin x$ s počáteční podmínkou $y(0) = 1$.

```
> DR:=diff(y(x),x)*y(x)^2+y(x)=sin(x);
```

$$DR := \left(\frac{d}{dx} y(x)\right) y(x)^2 + y(x) = \sin(x)$$

```
> dsolve({DR,y(0)=1},y(x));
```

Příkaz **dsolve** nám nedává řešení, úlohu je potřeba řešit numericky. Do příkazu dsolve zadáme parametr **numeric**.

```
> res:=dsolve({DR,y(0)=1},numeric);
```

$$res := \mathbf{proc}(x\_rkf45) \ \ldots \ \mathbf{end\ proc}$$

Maple úlohu vyřeší numericky pomocí metody Rungeho-Kutty a do res uloží proceduru. Chceme určit hodnotu řešení   např. v $x = 1$.

```
> res(1);
```

$$[x = 1., y(x) = 0.676752337623565]$$

Hodnota $y(x)$, kterou jsme obdrželi, je funkční hodnota v bodě $x = 1$.
Abychom získali představu o řešení, je možné si ho pomocí následujícího příkazu **odeplot** nechat vykreslit.
V nejjednodušším případě stačí do odeplot zapsat název funkce, kterou chceme vykreslit a společně s ní i interval, na kterém řešení hledáme. Protože příkaz odeplot funguje jen s balíčkem plots, je možné balíček použít jen pro tento případ a zapsat ho následujícím způsobem.

```
> plots[odeplot](res,0..1);
```



Příklad 2. Nalezněte řešení diferenciální rovnice z Příkladu 1. s použitím metody Rungeho-Kutty. Pro použití metody Rungeho-Kuttyje potřeba načíst balíček **Student[NumericalAnalysis].** Příkaz je tvaru **InitialValueProblem(DR, PP, x=b, opts),** kde **DR** je daná diferenciální rovnice zapsaná ve tvaru na kterém hledáme

Pro použití metody Rungeho-Kutty je potřeba načíst balíček **Student [NumericalAnalysis].** Příkaz je tvaru **InitialValueProblem(DR, PP, x=b, opts),** kde **DR** je daná diferenciální rovnice zapsaná ve tvaru $y' = f(y, x)$, **PP** je počáteční podmínka (počáteční bod intervalu, na kterém hledáme řešení), **x = b** je koncový bod intervalu, na kterém hledáme řešení, **opts** jsou nepovinné parametry. Pokud do opts nic nezapíšeme, použije Maple Eulerovu metodu. Pro metodu Rungeho-Kutty nastavíme **method=rungekutta.**

řešení, **opts**jsou nepovinné parametry. Pokud do opts nic nezapíšeme, použije Maple Eulerovu metodu. Pro metodu Rungeho-Kutty nastavíme **method=rungekutta.**
Další parametry opts jsou v HELPU. Pro nás je užitečné vědět, že pokud zadáme **output= plot**, získáme grafické řešení úlohy, pokud zadáme **output=information**, obdržíme tabulku, ve které je interval, na kterém hledáme řešení, rozdělen uzlovými body, v nich ž jsou vypočítané funkční hodnoty a absolutní chyby v těchto bodech oproti exaktnímu řešení. Zadáme-li **output=Error** získáme absolutní chybu v koncovém krajním bodě oproti exaktnímu řešení.
V Maplu je vnitřně nastaveno dělení intervalu, na kterém hledáme řešení, na 5 podintervalů. Pokud chceme toto dělení změnit, zadáme parametr **numsteps=n,**kde n je počet podintervalů.

```
> with(Student[NumericalAnalysis]):
```

```
> InitialValueProblem(diff(y(x),x)=(sin(x)-y(x))/y(x)^2,y(0)=1,x=1,
method=rungekutta);
```

0.6904

Hodnota, kterou jsme obdrželi, je $y(1)$, tj. funkční hodnota v koncovém krajním bodě intervalu, na kterém hledáme řešení.

```
> InitialValueProblem(diff(y(x),x)=(sin(x)-y(x))/y(x)^2,y(0)=1,x=1,
method=rungekutta,output=plot);
```



V grafu, který jsme získali, je exaktní řešení a řešení metodou RungeKuttyho.

```
> InitialValueProblem(diff(y(x),x)=(sin(x)-y(x))/y(x)^2,y(0)=1,x=1,
method=rungekutta,numsteps=10,output=plot);
```



Zvýšíme-li počet dělících bodů, řešení je přesnější.
Příklad 3. Nalezněte řešení diferenciální rovnice z Příkladu 1. s použitím zabudované funkce **Euler**.
Při použití funkce **Euler** budeme postupovat stejně jako u funkce RungeKutta.

```
> with(Student[NumericalAnalysis]):
> 
InitialValueProblem(diff(y(x),x)=(sin(x)-y(x))/y(x)^2,y(0)=1,x=1)
;
```

$$0.6605$$

Hodnota, kterou jsme obdrželi, je $y(1)$, tj. funkční hodnota v koncovém krajním bodě. Srovnejte, jakou hodnotu jsme získali za použití metody Rungeho-Kutty.

```
> 
InitialValueProblem(diff(y(x),x)=(sin(x)-y(x))/y(x)^2,y(0)=1,x=1,
output=Error);
```

$$0.01621$$

Absolutní chyba v koncovém krajním bodě (oproti exaktnímu řešení).

```
> 
InitialValueProblem(diff(y(x),x)=(sin(x)-y(x))/y(x)^2,y(0)=1,x=1,
output=plot);
```




```
> InitialValueProblem(diff(y(x),x)=(sin(x)-y(x))/y(x)^2,y(0)=1,x=1,
output=information);
```

x	$y(x)$	$[Euler]$	$[Error]$
0.	1.	1.	0.
0.2000	0.8038	0.8000	0.0038
0.4000	0.6428	0.6121	0.0307
0.6000	0.5774	0.4932	0.0842
0.8000	0.6089	0.5519	0.0570
1.	0.6768	0.6605	0.0163

Příklad 4. Srovnejte exaktní řešení a řešení Eulerovou metodou a metodou Rungeho-Kutty z Příkladu 1.

Pro srovnání použijeme příkaz InitialValueProblem, do něj zadáme diferenciální rovnici, počáteční podmínku, koncový bod intervalu, metodu, kterou úlohu řešíme, a metodu (nebo metody), se kterou děláme srovnání. Výstup volíme ve formě grafu nebo informační tabulky.

```
> InitialValueProblem(diff(y(x),x)=(sin(x)-y(x))/y(x)^2,y(0)=1,x=1,
method=rungekutta,submethod=rk4,
comparewith=[[euler]],output=plot);
```




```
> InitialValueProblem(diff(y(x),x)=(sin(x)-y(x))/y(x)^2,y(0)=1,x=1,method=rungekutta,submethod=rk4,comparewith=[[euler]],output=information);
```

x	$y(x)$	[R-K 4th Ord.]	[Error]	[Euler]	[Error]
0.	1.	1.	0.	1.	0.
0.2000	0.8038	0.8038	0.	0.8000	0.0038
0.4000	0.6428	0.6428	0.	0.6121	0.0307
0.6000	0.5774	0.5778	0.0004	0.4932	0.0842
0.8000	0.6089	0.6096	0.0007	0.5519	0.0570
1.	0.6768	0.6771	0.0003	0.6605	0.0163

>